

【時刻の表示】**【DateTime オブジェクトの作成】****サンプルコード**

```

||以下のコードはDateTime オブジェクトを作成するためのサンプルです
It is currently {DateTime}.

The first day of the 21st century is
{DateTime "01/01/2001"}.

The last day of the 20th century is
{DateTime year=2000, month=12, day=31}.

Golf was last played on the Moon
{DateTime "2/6/71 7:12:22 -0500"}.

```

実行結果

```

It is currently 2013-07-09 15:14:40.539999.

The first day of the 21st century is 2001-01-01 00:00:00.000000.

The last day of the 20th century is 2000-12-31 00:00:00.000000.

Golf was last played on the Moon 1971-02-06 07:12:22.000000 -0500.

```

【DateTime の文字列への変換】**サンプルコード**

```

||以下のコードはDateTime を文字列へ変換するためのサンプルです
{let moon-golf:DateTime =
  {DateTime "2/6/71 7:12:22 -0500"}
}
||実行結果を確認してください
Golf was last played on the moon on {value moon-golf.info.locale-date} at
{value moon-golf.info.locale-time}.
To be specific it was {value moon-golf}.

```

実行結果

```

Golf was last played on the moon on 1971/02/06 at 7:12:22. To be specific it was 1971-02-06
07:12:22.000000 -0500.

```

【日付と時刻の要素】

サンプルコード

```
||以下のコードはDateTimeInfo クラスを使ったサンプルです
{let now:DateTime = {DateTime}}

||実行結果を確認してください
Today is {value now.info.locale-full}
(that is,
 {value now.info.locale-weekday}
 the {ordinal now.info.day} day
 of the {ordinal now.info.month} month
 ({value now.info.locale-month})
 of the year {value now.info.year}
 at {value now.info.locale-time} {value now.info.timezone-name}).

It is the {ordinal now.info.day-of-year} day of the year, which
is the {ordinal now.info.day-of-week} day of the week, called
{value now.info.locale-weekday}.
```

実行結果

Today is 2013/07/09 15:14:40 (that is, 火曜日 the 9 day of the 7 month (7月) of the year 2013 at 15:14:40 東京 (標準時)).

It is the 190 day of the year, which is the 2 day of the week, called 火曜日.

【時間の長さ】

【時間の長さ】

サンプルコード

```
||以下のコードはTimeクラスを使ったサンプルです
{let some-seconds:Time = 23479233seconds}
{let some-minutes:Time = 232643minutes}
{let some-hours:Time = 83.343hours}
{let total-time:Time = some-seconds + some-minutes + some-hours}

||実行結果を確認してください
The total time is {value total-time / 1day} days.
```

実行結果

The total time is 436.781 days.

【DateTime の四則演算1】

サンプルコード

```
||以下のコードはDateTime の四則演算に関するサンプルです
{let now:DateTime = {DateTime}}
{let leave-in:Time = .5hour}

||実行結果を確認してください
If you want to leave in {value leave-in / 1s}
seconds, set your alarm for
{value (now + leave-in).info.locale-time}.
```

実行結果

If you want to leave in 1800 seconds, set your alarm for 15:44:40.

【DateTime の四則演算2】

サンプルコード

```
{let now:DateTime = {DateTime}}
```

```
||以下のコードはDateTimeクラスの部分的な加算減算を行うためのサンプルです  
{let returnday:DateTime = {DateTime now, day={value now.info.day + 80}}}
```

```
||実行結果を確認してください
```

```
If you left on an eighty-day around the world trip today,  
you'd get back on {value returnday.info.locale-date}
```

```
Midnight of the last day of the previous month was  
{value {DateTime day=0}.info.locale-full}
```

実行結果

```
If you want to leave in 1800 seconds, set your alarm for 15:44:40.
```

【経過時間】

サンプルコード

```
|| 以下のコードは経過時間を算出するサンプルです  
|| DateTime.elapsed メソッドにて実現できます  
{let midnight:DateTime = {DateTime hour=24}}  
  
|| 実行結果を確認してください  
It will be tomorrow in  
{value -{midnight.elapsed} / 1hour} hours.
```

実行結果

It will be tomorrow in 8.75541 hours.

【日付・時間の比較方法】

【日付と時刻の比較】

サンプルコード

```
||以下のコードは日付と時刻の比較に関するサンプルです
{let now:DateTime = {DateTime}}
{let breakfast:DateTime = {DateTime now, hour=7}}
{let lunch:DateTime = {DateTime now, hour=12}}
{let dinner:DateTime = {DateTime now, hour=20}}

||実行結果を確認してください
{if {breakfast.compare now} < 0 then
  {text You missed breakfast.}
else
  {text
    Breakfast is in
    {value -{breakfast.elapsed ending=now} / 1hour} hours.}
}
{if {lunch.compare now} < 0 then
  {text You missed lunch.}
else
  {text
    Lunch is in
    {value -{lunch.elapsed ending=now} / 1hour} hours.}
}
{if {dinner.compare now} < 0 then
  {text You missed dinner.}
else
  {text
    Dinner is in
    {value -{dinner.elapsed ending=now} / 1hour} hours.}
}
```

実行結果

You missed breakfast. You missed lunch. Dinner is in 4.75541 hours.